# $Id: PKGBUILD 137677 2011-09-10 11:50:26Z pierre $
# Maintainer: simo <simo@archlinux.org> 
# Contributor: Manolis Tzanidakis

pkgname=compface
pkgver=1.5.2
pkgrel=3
pkgdesc="Utils & libs to convert from/to X-Face format."
arch=('i686' 'x86_64')
url="http://ftp.xemacs.org/pub/xemacs/aux/"
license=('custom')
depends=('glibc')
source=("$url/$pkgname-$pkgver.tar.gz" 'LICENSE')
md5sums=('62f4f79c0861ad292ba3cf77b4c48319'
         '27a0d5a4d695e68b4d3eebe17db77189')

build() {
  cd $srcdir/$pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd $srcdir/$pkgname-$pkgver
  make BINDIR=$pkgdir/usr/bin LIBDIR=$pkgdir/usr/lib MANDIR=$pkgdir/usr/share/man \
    INCLUDEDIR=$pkgdir/usr/include install
  install -D -m644 ../LICENSE $pkgdir/usr/share/licenses/$pkgname/LICENSE
}
